TD1 - Sommes, produits et complexes

Nicolas Estibals*

14, 21 et 28 novembre 2008

Exercice 1

1. Vous savez certainement exprimer $\sum_{k=0}^{n} k$ et $\sum_{k=0}^{n} k^2$ en fonction de n, retrouvez ces expressions grâce à Maple.
2. Sauriez-vous calculer $\sum_{k=0}^{n} k^3$ en n'utilisant que les résultats précédents ?
   (Indication : Exprimez de deux manières différentes $\sum_{k=0}^{n}(k+1)^4 - k^4$)
3. Vérifiez le résultat grâce à Maple.
4. Calculez la somme des $f(i,j) = (i+j)^2$ dans le triangle représenté ci-contre.



Exercice 2

1. Mettre sous forme trigonométrique les deux complexes suivants :

$$\begin{cases} z_1 & = -1 - i \\ z_2 & = \frac{1-i\sqrt{3}}{2} \end{cases}$$

2. Déduire du produit de z_1 et z_2 la valeur de $\cos \frac{\Pi}{12}$.

Exercice 3 Indicatrice d'Euler

L'indicatrice φ d'Euler est la fonction qui associe à n le nombre d'entiers inférieur strictement à n, premiers avec n.

1. Écrire une procédure `ind_euler` qui teste tous les entiers inférieurs à son argument n et compte ceux qui sont premier avec n.
   (Indication : La fonction `igcd` donne le PGCD de deux entiers en Maple, regardez l'aide en ligne sur cette fonction avant de vous en servir.)
2. Écrire une fonction qui renvoie $(p-1) * p^{k-1}$ si elle reçoit en entrée la liste `[p,k]`.
3. On peut montrer que si $n = \prod_{i=1}^{q} p_i^{k_i}$ alors $\varphi(n) = \prod_{i=1}^{q} (p_i - 1) * p_i^{k_i - 1}$. Utilisez cette formule pour écrire une deuxième procédure qui calcul l'indicatrice d'Euler.
   (Indication : la fonction `ifactors` fait la décomposition en facteur premier.)
4. Tracer le graphe de φ sur les 10000 premiers entiers.

Exercice 4 Racine n-ième de l'unité

Les racines n-ième de l'unité sont les racines complexes du polynôme $X^n - 1$.

1. Définir la fonction de deux variables qui associe $e^{2ki\Pi/n}$ à (k,n).
2. Vérifier que $e^{2ki\Pi/n}$ est une racine n-ième.
3. Montrer que la somme des racines n-ième est nulle.
4. Exprimer simplement leur produit.

---

*N'hésitez pas à m'écrire à l'adresse `Nicolas.Estibals@ens-lyon.fr`!
Vous pouvez retrouver les sujets sur `http://perso.ens-lyon.fr/nicolas.estibals/`.

```
> # Exercice 1
restart; # N'oubliez jamais de faire un restart à chaque début d'exercice
         # pour éviter les mauvaises surprises.
```

```
> # Question 1
# Prenez l'habitude de docummenter vos feuilles  de calculs c'est plus agréable
# pour le correcteur et pour vous y retrouver.
sum(k,k=0..n); sum(k^2,k=0..n);
```

$$\frac{1}{2}(n+1)^2-\frac{1}{2}n-\frac{1}{2}$$

$$\frac{1}{3}(n+1)^3-\frac{1}{2}(n+1)^2+\frac{1}{6}n+\frac{1}{6}$$

```
> # Pour retrouver l'expression que vous avez apprise, il faut factoriser.
S1:=factor(sum(k,k=0..n));
S2:=factor(sum(k^2,k=0..n));
```

$$S1 := \frac{1}{2}n(n+1)$$

$$S2 := \frac{1}{6}n(n+1)(2n+1)$$

```
> # Question 2
# 'Sum' permet d'exprimer une somme sans la calculer contrairement à 'sum'.
Sum((k+1)^4-k^4,k=0..n);
```

$$\sum_{k=0}^{n}((k+1)^4-k^4)$$

```
> # Cette somme est égale à (n+1)^4 !
Sum((k+1)^4-k^4,k=0..n)=(n+1)^4;
#Attention ici Maple n'a pas vérifié l'égalité, c'est vous qui lui avez demandez de l'écrire!
```

$$\sum_{k=0}^{n}((k+1)^4-k^4)=(n+1)^4$$

```
> # Développons (k+1)^4-k^4
expand((k+1)^4-k^4);
```

$$4k^3+6k^2+4k+1$$

```
> # Nous avons donc:
Sum((k+1)^4-k^4,k=0..n) = 4*Sum(k^3,k=0..n) + 6*Sum(k^2,k=0..n) +
4*Sum(k,k=0..n) + Sum(1,k=0..n);
```

$$\sum_{k=0}^{n}((k+1)^4-k^4)=4\left(\sum_{k=0}^{n}k^3\right)+6\left(\sum_{k=0}^{n}k^2\right)+4\left(\sum_{k=0}^{n}k\right)+\left(\sum_{k=0}^{n}1\right)$$

```
> #En remplaçant on obtient :
Sum(k^3,k=0..n)=((n+1)^4 -6 * S2 -4*S1 -sum(1,k=0..n))/4;
```

$$\sum_{k=0}^{n}k^3=\frac{1}{4}(n+1)^4-\frac{1}{4}n(n+1)(2n+1)-\frac{1}{2}n(n+1)-\frac{1}{4}n-\frac{1}{4}$$

```
> factor(((n+1)^4 -6 * S2 -4*S1 - sum(1,k=0..n))/4);
```

$$\frac{1}{4}n^2(n+1)^2$$

```
> # Question 3
factor(sum(k^3,k=0..n));
```

$$\frac{1}{4}n^2(n+1)^2$$

```
> # Question 4
# La difficulté de cette question est de ne pas se tromper dans les indices des deux sommes
# imbriquées. Il ne faut pas hésiter à faire un dessin pour calculer les indices.
> sum(sum((i+j)^2,j=0..(n-1-i)),i=0..n-1);
```

$$\frac{1}{4}n^4-\frac{1}{6}n^3-\frac{1}{4}n^2+\frac{1}{6}n$$

```
> # Exercice 2
> restart;
> #Question 1
> #Définissons les dans un premier temps.
> z1:=-1-I;
> z2:=1/2-3^(1/2)/2*I;
```

$$z1 := -1 - I$$

$$z2 := \frac{1}{2} - \frac{1}{2} I \sqrt{3}$$

```
> #Il y a deux manières de faire :
> abs(z1); argument(z1);
> abs(z2); argument(z2);
```

$$\sqrt{2}$$

$$-\frac{3}{4}\pi$$

$$1$$

$$-\frac{1}{3}\pi$$

```
> # et de façon plus concise:
> Z1:=convert(z1,polar);
> Z2:=convert(z2,polar);
```

$$Z1 := \mathrm{polar}\left(\sqrt{2}, -\frac{3}{4}\pi\right)$$

$$Z2 := \mathrm{polar}\left(1, -\frac{1}{3}\pi\right)$$

```
> #Question 2
> #Nous pouvons calculer ce produit en coordonnées cartésiennes:
> expand(z1*z2);
> #ou polaire:
> simplify(Z1*Z2);
```

$$\left(-\frac{1}{2} - \frac{1}{2} I\right) + \frac{1}{2} I \sqrt{3} - \frac{1}{2}\sqrt{3}$$

$$\mathrm{polar}\left(\sqrt{2}, -\frac{13}{12}\pi\right)$$

```
> #En s'intéressant à la partie réel de ces deux nombres (qui sont égaux!)
> Re(expand(z1*z2));
> Re(simplify(Z1*Z2));
```

$$-\frac{1}{2} - \frac{1}{2}\sqrt{3}$$

$$-\sqrt{2}\cos\left(\frac{1}{12}\pi\right)$$

```
> #Et finalement cos(Pi/ 12) est donc:
> Re(expand(z1*z2))/(-sqrt(2));
> simplify(%);
>
```

$$-\frac{1}{2}\left(-\frac{1}{2} - \frac{1}{2}\sqrt{3}\right)\sqrt{2}$$

$$\frac{1}{4}(1 + \sqrt{3})\sqrt{2}$$

```
> # Exercice 3
> restart;
> # Question 1
> ind_euler := proc(n)
> local compteur, i; # Il faut déclarer toutes les variables locales.
> compteur:=0;
> for i from 1 to n-1 do
>         if (igcd(n,i)=1) then
>                 #Dans ce cas i est premier avec n.
>                 compteur:=compteur+1;
>         fi;
> od;
> return compteur;
  end;
   ind_euler := proc(n)
   local compteur, i;
       compteur := 0;
       for i to n - 1 do if igcd(n, i) = 1 then compteur := compteur + 1 end if; end do;
       return compteur;
   end proc;
> # Testons notre procédure
  ind_euler(8);
  ind_euler(17);
  ind_euler(144);
                                     4
                                     16
                                     48
> # Question 2
> # Si liste = [p, k], liste[1]=p et liste[2]=k.
> f := liste -> (liste[1] -1)* (liste[1]^(liste[2]-1));
```

$$f := liste \to (liste_1 - 1)\, liste_1^{(liste_2 - 1)}$$

```
> # Vérifions que nous ne nous sommes pas trompés.
> f([p,k]);
```

$$(p - 1)\, p^{(k - 1)}$$

```
> # Question 3
> # Commençons par un exemple.
> # 'ifactors(n)' renvoie le signe de n et la liste des [p_i,k_i]
> ifactors(120);
                          [1, [[2, 3], [3, 1], [5, 1]]]
> # On sélectionne la liste des [p_i,k_i]
> ifactors(120)[2];
                          [[2, 3], [3, 1], [5, 1]]
> # Utilisons 'map' et la fonction de la question 2 pour construire les
  facteurs du
> # produit à calculer.
> map(f,%);
                                 [4, 2, 4]
> # Il ne reste plus qu'a faire le produit :
> convert(%,`*`);
                                     32
> # Écrivons maintenant une belle procédure qui réalise toutes ces étapes :
> ind_euler2 := proc(n)
> local liste_facteurs_premiers, liste, resultat;
> liste_facteurs_premiers:= ifactors(n)[2];
> liste := map(f,liste_facteurs_premiers);
> resultat := convert(liste,`*`);
  return resultat;
> end;
          ind_euler2 := proc(n)
          local liste_facteurs_premiers, liste, resultat;
              liste_facteurs_premiers := ifactors(n)[2];
              liste := map(f, liste_facteurs_premiers);
              resultat := convert(liste, `*`);
              return resultat;
          end proc;
> # Testons notre procédure
  ind_euler2(8);
  ind_euler2(17);
  ind_euler2(144);
                                     4
                                     16
                                     48
> # Question 4
> # On va utiliser la commande 'plot'
  with(plots);
```

```
> plot([seq([n,ind_euler2(n)],n=1..10000)],style=point,symbol=cross,title='
  ndicatrice d'Euler");
```

Warning, the name changecoords has been redefined

[*animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, cylinderplot, densityplot, display, display3d, fieldplot, fieldplot3d, gradplot, gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra_supported, polyhedraplot, replot, rootlocus, semilogplot, setoptions, setoptions3d, spacecurve, sparsematrixplot, sphereplot, surfdata, textplot, textplot3d, tubeplot*]

Indicatrice d'Euler




```
> # Exercice 4
  restart;
> # Question 1
  # Notez que l'on peut définir des fonctions de deux variables.
  r:= (k,n) -> exp( 2*I*Pi*k/n);
```

$$r := (k, n) \rightarrow \mathbf{e}^{\left(\frac{2\, I\, \pi\, k}{n}\right)}$$

```
> # Question 2
  # On va montrer que cette expression est nulle
> r(k,n)^n-1;
```

$$\left(\mathbf{e}^{\left(\frac{2\, I\, \pi\, k}{n}\right)}\right)^n - 1$$

```
# Il faut dire à Maple que n et k sont des entiers.
assume(n::integer,k::integer);
> simplify(%);
```

$$0$$

```
> # Question 3
  sum(r(k,n),k=0..n-1);
```

$$0$$

```
> # Question 4
> product(r(k,n),k=0..n-1);
```

$$(-1)^{(n\sim - 1)}$$